Math 4530

Friday April 20

Christoffel symbols and Gauss' Theorem Egregium

```
> restart:
    #always a good idea
> with(linalg):
    #as much as possible we will use matrix algebra
Warning, the protected names norm and trace have been redefined and unprotected
> #dot product
  dp:=proc(X,Y)
  X[1]*Y[1]+X[2]*Y[2]+X[3]*Y[3];
  end:
> nrm:=proc(X)
  sqrt(dp(X,X));
  end:
> #cross product:
  xp:=proc(X,Y)
  local a,b,c;
  a:=X[2]*Y[3]-X[3]*Y[2];
  b:=X[3]*Y[1]-X[1]*Y[3];
  c:=X[1]*Y[2]-X[2]*Y[1];
  [a,b,c];
  end:
> #Derivative matrix for mapping X:
  DXq:=proc(X)
  local Xu,Xv;
  Xu:=matrix(3,1,[diff(X[1],u),diff(X[2],u),diff(X[3],u)]);
  Xv:=matrix(3,1,[diff(X[1],v),diff(X[2],v),diff(X[3],v)]);
  simplify(augment(Xu,Xv));
  end:
> #Matrix of first fundamental form:
  gij:=proc(X)
  local Y;
  Y:=evalm(DXq(X));
  simplify(evalm(transpose(Y)&*Y));
  end:
> #unit normal:
  N:=proc(X)
  local Y,Z,s;
  Y:=DXq(X);
  Z:=xp(col(Y,1),col(Y,2));
  s:=nrm(Z);
  simplify(evalm((1/s)*Z));
  end:
> #matrix of second fundamental form:
  hij:=proc(X)
```

```
local Y,Xu,Xv,Xuu,Xuv,Xvv,U,h11,h12,h22;
Y:=DXq(X);
U:=N(X);
Xu:=col(Y,1);
Xv:=col(Y,2);
Xuu:=[diff(Xu[1],u),diff(Xu[2],u),diff(Xu[3],u)];
Xuv:=[diff(Xu[1],v),diff(Xu[2],v),diff(Xu[3],v)];
Xvv:=[diff(Xv[1],v),diff(Xv[2],v),diff(Xv[3],v)];
h11:=dp(Xuu,U);
h12:=dp(Xuv,U);
h22:=dp(Xvv,U);
simplify(matrix(2,2,[h11,h12,h12,h22]));
end:

> #matrix A of the (opposite) of the differential
#of the normal map
aij:=proc(X)
local Y,H,G;
H:=hij(X);
G:=gij(X);
simplify(evalm(inverse(G)&*H));
end:
> #Gauss curvature from second fundamental form
GK2:=proc(X)
local A;
A:=aij(X);
det(A);
end:
> #Mean curvature
MK:=proc(X)
local A;
A:=aij(X);
(1/2)*trace(A);
end:
> #Christoffel symbols
Gamma:=proc(X)
local Gammas,G,Ginv,DG,i,j,k,l;
G:=X;
Ginv:=evalm(inverse(G));
DG:=array(1..2,1..2,1..2);
   DG[1,1,1]:=diff(G[1,1],u);
   DG[1,1,2]:=diff(G[1,1],v);
   DG[1,2,1]:=diff(G[1,2],u);
   DG[2,1,1]:=DG[1,2,1];
   DG[1,2,2]:=diff(G[1,2],v);
   DG[2,1,2]:=DG[1,2,2];
   DG[2,2,1]:=diff(G[2,2],u);
   DG[2,2,2]:=diff(G[2,2],v);
Gammas:=array(1..2,1..2,1..2);
```

```
    for i from 1 to 2 do
       for j from 1 to 2 do
          for l from 1 to 2 do
          Gammas[i,j,l]:=sum((1/2)*Ginv[k,l]*
            (DG[i,k,j]+DG[k,j,i]-DG[i,j,k]),k=1..2);
          od:
        od:
      od:
simplify(Gammas);
end:
```

```
> #derivatives of Christoffel symbols
DGamma:=proc(X)
local G,Gammas,DGammas,i,j,k,m;
G:=X;
Gammas:=Gamma(G);
DGammas:=array(1..2,1..2,1..2,1..2);
for i from 1 to 2 do
  for j from 1 to 2 do
    for m from 1 to 2 do
      DGammas[i,j,m,1]:=diff(Gammas[i,j,m],u);
      DGammas[i,j,m,2]:=diff(Gammas[i,j,m],v);
    od:
   od:
  od:
simplify(DGammas);
end:
```

```
> #Gauss curvature from first fundamental form
GK1:=proc(X)
local G,Gammas,DGammas,Ginv,K,i,r;
G:=X;
Ginv:=inverse(G);
Gammas:=Gamma(G);
DGammas:=DGamma(G);

  K:=sum(Ginv[1,i]*(DGammas[i,1,2,2] - DGammas[i,2,2,1]
        + sum(
           Gammas[i,1,r]*Gammas[r,2,2] -
Gammas[i,2,r]*Gammas[r,1,2],
          r=1..2)),
       i=1..2);


simplify(K);
end:
```

```
>
```

```
>
> test:=(u,v)->[u,v,k1*u^2+k2*v^2];
    #good idea to check our derivation on a graph:
```

$$test := (u, v) \to [u, v, k1\, u^2 + k2\, v^2]$$

```
> GK1(gij(test(u,v)));
    #from the first fundamental form
```

$$4\,\frac{k1\, k2}{(1+4\,k2^2\,v^2+4\,k1^2\,u^2)^2}$$

```
> GK2(test(u,v));
    #from the differential of the normal map
```

$$4\,\frac{k1\, k2}{(1+4\,k2^2\,v^2+4\,k1^2\,u^2)^2}$$

```
> stinv:=(u,v)->[2*u/(u^2+v^2+1),2*v/(u^2+v^2+1),(u^2+v^2-1)/(u^2+v^
  2+1)];
     #inverse of stereographic projection onto the sphere
```

$$stinv := (u, v) \to \left[2\,\frac{u}{u^2+v^2+1},\, 2\,\frac{v}{u^2+v^2+1},\, \frac{u^2+v^2-1}{u^2+v^2+1}\right]$$

```
> GK1(gij(stinv(u,v)));
  GK2(stinv(u,v));
     #notice the second expression also equals 1; it would’ve been
     #neater if I’d assumed v,u real.
```

$$1$$

$$\mathrm{csgn}\Big(\big(\overline{(u^2+v^2)}^{\,2}\,u^8+4\,\overline{(u^2+v^2)}^{\,2}\,u^6\,v^2+6\,\overline{(u^2+v^2)}^{\,2}\,u^4\,v^4+4\,\overline{(u^2+v^2)}^{\,2}\,u^2\,v^6+\overline{(u^2+v^2)}^{\,2}\,v^8$$
$$+2\left|u^2+v^2\right|^6 u^2+2\left|u^2+v^2\right|^6 v^2+\left|u^2+v^2\right|^8\big)\,\overline{(u^2+v^2)}^{\,2}\,{}^2\Big)$$

```
> helcat:=(u,v)->[cos(t)*sinh(v)*sin(u)+sin(t)*cosh(v)*cos(u),
    -cos(t)*sinh(v)*cos(u)+sin(t)*cosh(v)*sin(u),u*cos(t)+v*sin(t)];
  #in your homework you showed these are all isometric as the
  parameter
  #t varies, so Gauss curvature should be t-independent
```

$$helcat := (u, v) \to [\cos(t)\sinh(v)\sin(u)+\sin(t)\cosh(v)\cos(u),$$
$$-\cos(t)\sinh(v)\cos(u)+\sin(t)\cosh(v)\sin(u),\, u\cos(t)+v\sin(t)]$$

```
> GK1(gij(helcat(u,v)));
```

$$-\frac{1}{\cosh(v)^4}$$

```
> GK2(helcat(u,v));
```

$$-\frac{\mathrm{csgn}(\cosh(v)^2)^2\,(\sin(t)^2+\cos(t)^2)}{\cosh(v)^4}$$

```
> hypgij:=(u,v)->matrix(2,2,[1/v^2,0,0,1/v^2]);
  #the hyperbolic plane is the upper half plane, with conformal
  metric
```

```
#in which g11=g22=1/v^2.  This "surface" is not realizable as a
surface
#in R^3, yet it has a "curvature":
```

$$hypgij := (u, v) \rightarrow \mathrm{matrix}\left(2, 2, \left[\frac{1}{v^2}, 0, 0, \frac{1}{v^2}\right]\right)$$

```
> GK1(hypgij(u,v));
```

$$-1$$

```
>
```